

Integration of leJOS in die Eclipse (Helios)

Aegidius Pluess, Version 1.0, September 23, 2011

Requirements (example, folders could be named otherwise):

- Windows with path to *c:\bat*
- leJOS in folder *e:\Vejos*, path to *e:\Vejos\bin*, environment variable *NXJ_HOME* with value *e:\Vejos*

Create (or download) the following 3 batch files:

lejoslink.bat:

```
@echo off

rem Version 1.0 (www.aplu.ch)
rem Calls nxjlink.bat as Eclipse external tool
rem Assumptions:
rem - NXJ_HOME environment variable defined
rem - lejos_home\bin part of system path

if .%2 == . goto usage
if not .%3 == . goto usage

echo leJOS linking now...

set home=%1
set file=%2
echo Project home = %home%
echo Selected Java source file = %file%

rem Strip file extension
set classname=%file:~0,-5%
echo Java class name = %classname%

rem Call leJOS nxjlink from directory of .class file
cd /d %1\bin
call nxjlink %classname% -o%classname%.nxj

echo Linking finished.
goto quit

:usage
echo Usage: lejoslink ^<projectpath^> ^<sourcefile^>.java

:quit
```

lejosupload.bat:

```
@echo off

rem lejosupload.bat
```

```

rem Version 1.0 (www.aplu.ch)
rem Calls nxjupload.bat as Eclipse external tool
rem Assumptions:
rem - NXJ_HOME environment variable defined
rem - lejos_home\bin part of system path

if .%2 == . goto usage
if not .%4 == . goto usage

echo leJOS uploading now...

set home=%1
set file=%2
set bluetoothname=%3

echo Project home = %home%
echo Selected Java source file = %file%
echo Using Bluetooth name = %bluetoothname%

rem Strip file extension
set classname=%file:~0,-5%
echo Java class name = %classname%

rem Call leJOS nxjupload from directory of .class file
cd /d %1\bin
call nxjupload -r -b -n %bluetoothname% %classname%.nxj

rem Remove -b and -n parameter if USB upload is needed
rem call nxjupload -r %classname%.nxj

echo Uploading finished.
goto quit

:usage
echo Usage: lejosupload ^<projectpath^> ^<sourcefile^>.java ^<bluetoothname^>

:quit

```

lejoslinkandupload.bat:

```

@echo off

rem lejoslinkandupload.bat
rem Version 1.0 (www.aplu.ch)
rem Calls nxjlink.bat and nxjupload.bat as Eclipse external tool
rem Assumptions:
rem - NXJ_HOME environment variable defined
rem - lejos_home\bin part of system path

if .%2 == . goto usage
if not .%4 == . goto usage

```

```

call lejoslink.bat %1 %2
call lejosupload.bat %1 %2 %3
goto quit

:usage
  echo Usage: lejoslinkandupload ^<projectpath^> ^<sourcefile^>.java ^<blue
toothname^>

:quit

```

Adapt in *lejosupload.bat* the line containing *nxjupload* to your environment (modify NXT Bluetooth name, omit parameters *-b* and *-n*, if you use an USB connection).

As you see the 3 batch files looks for 3 parameters: the home directory of the project, the file name (including *.java*) of the Java source file and the NXT bluetooth name (if needed). These parameters can be generated automatically by Eclipse using the **External Tool** option.

Proceed as follows:

1. Start Eclipse
2. Menu *Run | External Tools | External Tools Configuration*
3. In the left window right-click *Program* and select *New*
4. In right window enter
Name: *LejosLinkAndUpload* (oder whatever you want)
Location: *c:\bat\lejoslinkandupload.bat*
Arguments:
`${project_loc} ${selected_resource_name} NXT`
where *NXT* is the Bluetooth name of your NXT
4. Click *Apply* (clicking *Run* generates an error message because *selected_resource_name* is undefined)
5. Click tab *Common*. In *Display in favorites menu* select the checkbox *External Tools*. Click *Apply*

Now you create a new Java project as usual. Add the leJOS library *classes.jar* as external jar. If you use the library *NxtJLibA* downloaded from <http://www.aplu.ch/nxt> the original *classes.jar* that resides in a subdirectory of the lejos-home directory must be **replaced** by the distributed *classes.jar* (backup the original file).

In the Package Explorer click your source file to select it. Saving it (*Ctrl-S*) compiles the source (you find the class file in the *bin* subfolder of the project root folder). Click in menu *Run | External Tool* the entry *LejosLinkAndUpload* and the processing of the batch files should be displayed in the Eclipse console window. **Keep in mind: the source file to compile, link and upload must always be selected in the Package Explorer, before the the External Tool is started!**

You may further simplify the process by creating a **shortcut** that executes the *Last External Tool* like this:

Menu *Window | Preferences*. In the left window open *General* and *Keys*. Enter in the *Filter* field *external* and select the command *Run Last Lauched External Tool*. Select field *Binding*: and type the shortcut you want to use, e.g. *Ctrl-T*

Supplement:

You can use the "**Quick Access**" option of Eclipse to execute a command. To define the command enter *Ctrl-3* and start to write *external*. In the displayed menu select *External Tools - Run LejosLinkAndUpload*. From now on you can press **Ctrl-3** to compile/link and upload your program. If you changed the source code, press the required sequence is **Ctrl-3, enter, enter**, because the source file has to be saved first.

As before the action is displayed in the Eclipse Console window. You can even have several source files in the same project. Select in the Package Explorer the source you want to use and press *Ctrl-3*.

Appendix:

To test the batch files without Eclipse, execute them with adequate parameters directly in a command shell.

Get the batch files from <http://www.aplu.ch/home/download/eclipsebatch.zip>

Comments to support@aplu.ch are welcome!